import crafttweaker.item.IIngredient;
import crafttweaker.item.IItemStack;
import crafttweaker.oredict.IOreDict;
import crafttweaker.oredict.IOreDictEntry;

#packmode normal simplified

print(" =================== Creative Items =================== ");

	// ATM Tweaks items vars
	var atmAnima            = <atmtweaks:item_material:4>;
	var atmOrb              = <atmtweaks:item_material:5>;
	var atmIndustrium       = <atmtweaks:item_material:2>;
	var atmForgedIndustrium = <atmtweaks:item_material:3>;
	var atmSphere           = <atmtweaks:item_material:6>;
	var atmDense            = <atmtweaks:item_material:8>;
	var atmUnifiedCore      = <atmtweaks:item_material:7>;
	var atmStar             = <atmtweaks:item_material>;

	var manaTabFull = <botania:manatablet>.withTag({mana: 500000}).onlyWithTag({mana:500000});
	
	//ATM Star
	recipes.addShapeless(atmStar, [atmOrb,atmUnifiedCore,atmForgedIndustrium,manaTabFull,manaTabFull,manaTabFull]);

	var uuIC2 = <ic2:misc_resource:3>;
	
	// atmUnifiedCore
	recipes.addShaped(atmUnifiedCore, [
			[uuIC2, <appliedenergistics2:material:48>, uuIC2],
			[uuIC2, <ic2:ingot>, uuIC2],
			[uuIC2, <rftoolsdim:dimlet_control_circuit:6>, uuIC2]
			]);
			
	// atmForgedIndustrium
	recipes.addShaped(atmForgedIndustrium, [
			[<minecraft:diamond_block>, <minecraft:diamond_block>, <minecraft:diamond_block>],
			[<ic2:ingot>, 				<ic2:ingot>, 			   <ic2:ingot>],
			[<minecraft:emerald_block>, <minecraft:emerald_block>, <minecraft:emerald_block>]
			]);
			
	// atmOrb
	recipes.addShaped(atmOrb, [
			[<extrautils2:opinium:8>, <extrautils2:opinium:8>, 		<extrautils2:opinium:8>],
			[<extrautils2:opinium:8>, <extrautils2:ingredients:5>,	<extrautils2:opinium:8>],
			[<extrautils2:opinium:8>, <extrautils2:opinium:8>, 		<extrautils2:opinium:8>]
			]);
	
print(" =================== Creative Items Use ATM =============== ");	
	
	recipes.addShaped(<storagedrawers:upgrade_creative>, [
		[null, <storagedrawers:upgrade_storage:4>, null],
		[<storagedrawers:upgrade_storage:4>, <atmtweaks:item_material>, <storagedrawers:upgrade_storage:4>],
		[null, <storagedrawers:upgrade_storage:4>, null]
	]);
	
	recipes.addShaped(<appliedenergistics2:creative_energy_cell>, [
		[<appliedenergistics2:dense_energy_cell>, null, <appliedenergistics2:dense_energy_cell>],
		[null, atmStar, null], 
		[<appliedenergistics2:dense_energy_cell>, null, <appliedenergistics2:dense_energy_cell>]
	]);
	
	// PSI 加入异常
	// recipes.addShaped(<psi:cad_assembly:5>, [
	// 	[<psi:cad_assembly:3>, null, <psi:cad_assembly:4>],
	// 	[null, atmStar, null], 
	// 	[<psi:cad_assembly:4>, null, <psi:cad_assembly:3>]
	// ]);
	
	recipes.addShaped(<botania:pool:1>, [
		[<botania:pool>, null, <botania:pool>],
		[null, atmStar, null], 
		[<botania:pool>, null, <botania:pool>]
	]);


	recipes.addShaped(<ic2:te:86>, [
		[<ic2:te:75>, null, <ic2:te:75>],
		[null, atmStar, null], 
		[<ic2:te:75>, null, <ic2:te:75>]
	]);
	
	recipes.addShaped(<extrautils2:creativeenergy>, [
		[<extrautils2:rainbowgenerator>, null, <extrautils2:rainbowgenerator>],
		[null, atmStar, null], 
		[<extrautils2:rainbowgenerator>, null, <extrautils2:rainbowgenerator>]
	]);
	
	recipes.addShaped(<draconicevolution:creative_rf_source>, [
		[<draconicevolution:reactor_core>, null, <draconicevolution:reactor_core>],
		[null, atmStar, null], 
		[<draconicevolution:reactor_core>, null, <draconicevolution:reactor_core>]
	]);

	
	// Mek Energy Cube
	var creativeMekCube = <mekanism:energycube>.withTag({tier: 4, mekData: {energyStored: 1.7976931348623157E308}});
	recipes.addShapedMirrored("CreativeCube", creativeMekCube, [
		[<mekanism:machineblock2:13>, <mekanism:energycube>, <mekanism:machineblock2:13>],
		[<mekanism:energycube>, 	  atmStar, 				 <mekanism:energycube>],
		[<mekanism:machineblock2:13>, <mekanism:energycube>, <mekanism:machineblock2:13>]
	]);
	
	
	mods.immersiveengineering.ArcFurnace.addRecipe(<avaritia:infinity_pickaxe>, 
		uuIC2, null, 30000, 20480,
		[<minecraft:stick>, <minecraft:stick>, <minecraft:emerald_block>, <minecraft:emerald_block>, <minecraft:emerald_block>]);

	mods.immersiveengineering.ArcFurnace.addRecipe(<avaritia:infinity_sword>, 
		uuIC2, null, 30000, 20480,
		[<minecraft:diamond_block>, <minecraft:emerald_block>, <minecraft:emerald_block>, <minecraft:emerald_block>]);	
	
	var GutoDiamondPickaxe = <minecraft:diamond_pickaxe>.withTag({ench: [{lvl: 125, id: 0}, {lvl: 125, id: 1}, {lvl: 125, id: 2}, {lvl: 125, id: 3}, {lvl: 125, id: 4}, {lvl: 5, id: 5}, {lvl: 5, id: 6}, {lvl: 125, id: 7}, {lvl: 125, id: 16}, {lvl: 125, id: 17}, {lvl: 125, id: 18}, {lvl: 125, id: 19}, {lvl: 125, id: 20}, {lvl: 125, id: 21}, {lvl: 5, id: 32}, {lvl: 125, id: 34}, {lvl: 125, id: 35}, {lvl: 5, id: 70}], Unbreakable: 1});
	
	recipes.addShapedMirrored("CreativeDiamondPickaxe", GutoDiamondPickaxe, [
		[uuIC2, <avaritia:resource>, uuIC2],
		[<minecraft:diamond_pickaxe>, atmStar, <minecraft:diamond_pickaxe>], 
		[uuIC2, <minecraft:diamond_block>, uuIC2]
	]);
	
	recipes.remove(<minecraft:diamond_pickaxe>);
	
	recipes.addShaped(<minecraft:diamond_pickaxe>, [
		[<minecraft:diamond_block>, <minecraft:diamond_block>, <minecraft:diamond_block>],
		[null, <minecraft:stick>, null],
		[null, <minecraft:stick>, null]
	]);
	

print(" =================== Creative Items End ===============");